
Abstract

The roundrect macros for METAPOST provide extremely configurable, ex-
tremely versatile rectangles (including rounded corners), intended primar-
ily for inclusion in documents produced by TEX and friends. The idea
was to provide a METAPOST-based replacement for the incredibly versa-
tile tcolorbox package; the macros are far from achieving that goal. But
they are nevertheless extremely useful.

1 Introduction

While TikZ and its many accompanying packages, particularly tcolorbox, are
wonderful and powerful tools, whenever using them I inevitably feel completely
lost, and I exert great effort doing comparatively simple things. Contrariwise,
thanks to my experience with the drm and dozenal packages, writing in META-
POST is quite straightforward for me. So I decided to try to write some generalized
macros to provide functionality similar to that of tcolorbox. It’s not even close
to that kind of flexibility or power, but it’s still quite useful and versatile, so I
make it available for anyone who might be interested.

This document was typeset in accordance with the docstrip utility, which
allows the automatic extraction of code and documentation from the same docu-
ment.

2 Prerequisites and Conventions

Some prerequesites for using this package are METAPOST itself (obviously). If
you’re using the package with LATEX, the gmp package would probably be helpful;
be sure to use the latex package option. Finally, the package internally calls
TEX.mp, so that is also required. All of these should be packaged in any reasonably
modern LATEX system, such as TEXLive or MikTEX.

This documentation assumes nothing about your personal TEX or METAPOST

environment. ConTEXt and the various forms of LuaTEX have METAPOST built-
in; with pdfLATEX, the author’s choice, one can use the gmp package to include the
source directly in one’s document (that’s what’s been done in this documentation)
or develop a simple script to compile them afterwards and include them in the
source via \includegraphics (probably the quickest option, since compilation is
done in advance). Here, we simply post the plain vanilla METAPOST code, and
let you work out those details however you prefer.

3 Basic Usage

The core of all the action is the roundrect macro; this will set up your rounded
rectangle in the plainest way possible. The first argument is the box’s height, the
second its width, and the third its name, by which you will draw it later:

roundrect(1in,2in)(rectangle);

draw rectangle;

We have some options with this. If you want a title bar in your box, say
rrtitlebar := true;:rrtitlebar

rrtitlebar := true;

roundrect(1in,2in)(rectangle);

draw rectangle;

By default, roundrect displays a lower border for the title bar; if you don’t
like this, say rrtitlebotborder := false;:rrtitlebotborder

rrtitlebar := true;

rrtitlebotborder := false;

roundrect(1in,2in)(rectangle);

draw rectangle;

The command is called rrtitlebotborder because it’s usually on the bottom
of the title bar; however, it retains this name even when the title bar is elsewhere.

Although it’s impossible to tell whether this box still has a title bar, we
can verify it (if we don’t believe it) by changing the background color with
rrtitlebgcolor:rrtitlebgcolor

rrtitlebar := true;

rrtitlebotborder := false;

rrtitlebgcolor := green;

roundrect(1in,2in)(rectangle);

draw rectangle;

And while we’re at it, we may as well color the background of the main part
of the box with rrinnercolor:rrinnercolor

rrtitlebar := true;

rrtitlebotborder := false;

rrinnercolor := blue;

rrtitlebgcolor := green;

roundrect(1in,2in)(rectangle);

draw rectangle;

We can also change the colors of the border around the box (rrbordercolor),rrbordercolor

the colors of the title bar’s border (rrtitlebordercolor), and the line betweenrrtitlebordercolor

the title bar and the body, if any (rrtitlebotbordercolor):rrtitlebotbordercolor

rrtitlebar := true;

rrtitlebotborder := false;

rrinnercolor := blue;

rrbordercolor := (0,1,1);

rrtitlebordercolor := (1,1,0);

rrtitlebotbordercolor := red;

rrtitlebotbordercolor := red;

rrtitlebgcolor := green;

roundrect(1in,2in)(rectangle);

draw rectangle;

If there is no title bar, then rrinnercolor and rrbordercolor will affect the
entire box:

rrinnercolor := red;

rrbordercolor := blue;

roundrect(1in,2in)(rectangle);

draw rectangle;

We can put text in the title bar with rrtitletext (which is colored byrrtitletext

rrtitlecolor) and rrbodytext (which is colored by rrbodytextcolor):rrtitlecolor

rrbodytext

rrtextcolor

rrinnercolor := red;

rrtitlebgcolor := blue;

rrtitlecolor := black;

rrtextcolor := white;

rrbodytext := "Checking to see what happens when lines wrap";

rrtitletext := "TITLE";

roundrect(1in,2in)(rectangle);

draw rectangle;

TITLE
Checking to see what hap-
pens when lines wrap

We see we should raise the title text; the default value is 1
6 of the box’s height,

which is too large for this small box. So we adjust it with rrtitleht; this isrrtitleht

expressed as a distance up from the bottom border of the title bar:

rrinnercolor := red;

rrtitlebgcolor := blue;

rrtitlecolor := black;

rrtextcolor := white;

rrbodytext := "Checking to see what happens when lines wrap";

rrtitletext := "TITLE";

rrtitleht := 0.3in;

roundrect(1in,2in)(rectangle);

draw rectangle;

TITLE
Checking to see what hap-
pens when lines wrap

Plainly, this is terrible style; but the principles are accurately shown here.
We can easily change the font separately for both title and body with

rrtitlefont and , with the usual font-changing commands that we would userrtitlefont

rrtextfont in our text. We’ve also been using rectangles with extremely rounded borders;
this reflects the default “border radius” of roundrect, which is 40pt. We can
change how rounded our corners are with rrborderrad, and how wide our borderrrborderrad

is with rrborderwd:rrborderwd

rrtitlebar := true;

rrborderrad := 20pt;

rrborderwd := 5pt;

rrbodytext := "Checking to see what happens when lines wrap";

rrtitletext := "TITLE";

rrtitlefont := "\fontsize{14pt}{17pt}\selectfont\ bfseries\ sffamily";

rrtextfont := "\fontsize{14pt}{17pt}\selectfont\itshape";
rrtitleht := 0.3in;

roundrect(1in,2in)(rectangle);

draw rectangle;

TITLE
Checking to see what
happens when lines
wrap

Be sure, if you are calling METAPOST from LATEX via gmp or some similar
package, that you wrap any LATEX commands in an \unexpanded environment;
otherwise LATEX will expand them in a way that METAPOST won’t be able to
process.

We can even tell roundrect that we want our title bar on the bottom, with
rrtitleside, or aligned right or left rather than centered, with rrtitlepos:rrtitleside

rrtitlepos

6



rrtitlebar := true;

rrbodytext := "Checking to see what happens when lines wrap a lot; will it overlap the title bar or not";

rrtitletext := "TITLE";

rrtitletext := "TITLE";

rrtitleht := 0.3in;

rrtitleside := "bot";

rrtitlepos := "right";

roundrect(1in,2in)(rectangle);

draw rectangle;

TITLE

Checking to see what hap-
pens when lines wrap a lot;
will it overlap the title bar
or not

At present, the only options for rrtitleside are “top” and “bot”; “right” and
“left” should be added soon, once the code is refactored to make it easier. The
options for rrtitlepos are, predictably, “left”, “right”, and “center”.

To restore all these values to the defaults, simply issue the directive rrrestorevals;;rrrestorevals

this macro does nothing but replace the default values in all these parameters.
And those are the roundrect macros; I hope they are useful for someone.

Happy METAPOSTing!

4 Further Work Needed

This really boils down to just one thing: code refactoring. There’s too much dupli-
cation here; that’s what made it hard to add in different values for rrtitleside,
where it should have been quite easy. At the moment, I don’t really have time,
so I just put this together as it was and sent it out; but once I move some other
projects off my plate, this will get done. But for those who look at the code here
and are disgusted by the unnecessary duplication: I know, and I want to fix it.

5 Implementation

1 input TEX;

2 color rrinnercolor; rrinnercolor := white;

3 color rrbordercolor; rrbordercolor := black;

4 numeric rrborderwd; rrborderwd := 2pt;

5 numeric rrborderrad; rrborderrad := 40pt;

6 string rrtitlefont; rrtitlefont := "\fontsize{14pt}{17pt}\selectfont ";

7 string rrtextfont; rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";

8 color rrtitlecolor; rrtitlecolor := black;

9 color rrtextcolor; rrtextcolor := black;

10 color rrtitlebgcolor; rrtitlebgcolor := white;

11 color rrtitlebordercolor; rrtitlebordercolor := black;

12 boolean rrtitlebar; rrtitlebar := false;

13 string rrtitleside; rrtitleside := "top";

14 numeric rrtitleht; rrtitleht := 0pt;

15 string rrtitlepos; rrtitlepos := "center";

16 boolean rrtitlebotborder; rrtitlebotborder := true;

17 color rrtitlebotbordercolor; rrtitlebotbordercolor := black;

18 string rrtitletext; rrtitletext := "";

19 string rrbodytext; rrbodytext := "";

20 string rrbodyalign; rrbodyalign := "justify";

21 string rrbodyaligncom; rrbodyaligncom := "";

22 def rrrestorevals =

23 rrinnercolor := white;

24 rrbordercolor := black;

25 rrborderwd := 2pt;

26 rrborderrad := 40pt;

27 rrtitlefont := "\fontsize{14pt}{17pt}\selectfont ";

28 rrtextfont := "\fontsize{10pt}{12pt}\selectfont ";

29 rrtitlecolor := black;

30 rrtextcolor := black;

31 rrtitlebgcolor := white;

32 rrtitlebordercolor := black;

33 rrtitlebar := false;

34 rrtitleside := "top";

35 rrtitleht := 0pt;

36 rrtitlepos := "center";

37 rrtitlebotborder := true;

38 rrtitlebotbordercolor := black;

39 rrtitletext := "";

40 rrbodytext := "";

41 rrbodyalign := "justify";

42 rrbodyaligncom := "";

43 enddef;

44 def roundrect(expr rrht, rrwd)(suffix name) =

45 TEXPRE("%&latex" & char(10) & "\documentclass{article}\begin{document}");

46 TEXPOST("\end{document}");

47 if (rrtitleht = 0pt):

48 rrtitleht := rrht/6;

49 fi

50 path rra; path rrb; path rrc; path rrd;

51 pair a; pair b; pair c; pair d;

52 a := (0,0) shifted (-rrwd/2,-rrht/2);

53 b := (0,0) shifted (rrwd/2,-rrht/2);

54 c := (0,0) shifted (rrwd/2,rrht/2);

55 d := (0,0) shifted (-rrwd/2,rrht/2);

56 rra := fullcircle scaled rrborderrad shifted (xpart a +

57 (rrborderrad/2),ypart a + (rrborderrad/2));

58 rrb := fullcircle scaled rrborderrad shifted (xpart b -

59 (rrborderrad/2),ypart b + (rrborderrad/2));

60 rrd := fullcircle scaled rrborderrad shifted (xpart d +

61 (rrborderrad/2),ypart d - (rrborderrad/2));

62 rrc := fullcircle scaled rrborderrad shifted (xpart c -

63 (rrborderrad/2),ypart c - (rrborderrad/2));

64 pair f; f := (a--b) intersectionpoint rra;

65 pair g; g := (a--b) intersectionpoint rrb;

66 pair h; h := (b--c) intersectionpoint rrb;

67 pair i; i := (b--c) intersectionpoint rrc;

68 pair j; j := (c--d) intersectionpoint rrc;

69 pair k; k := (c--d) intersectionpoint rrd;

70 pair l; l := (d--a) intersectionpoint rrd;

71 pair m; m := (d--a) intersectionpoint rra;

72 picture name;

73 picture border;

74 pair n; pair o; path rrtitlepath; path rrfinalline;

75 name := image(fill f--g{right}..{up}h--i{up}..{left}j--

76 k{left}..{down}l--m{down}..{right}f--cycle

77 withcolor rrinnercolor);

78 border := image(pickup pencircle scaled rrborderwd;

79 draw f--g{right}..{up}h--i{up}..{left}j--

80 k{left}..{down}l--m{down}..{right}f--cycle

81 withcolor rrbordercolor);

82 addto name also border;

83 pair rrtitlestation;

84 picture rrtitlelabel;

85 if (rrtitlebar):

86 if (rrtitleside = "top"):

87 n := (xpart m,ypart k - (rrtitleht));

88 o := (xpart i,ypart k - (rrtitleht));

89 defaultscale := 4;

90 if (ypart n > ypart l):

91 n := (xpart n,ypart l);

92 o := (xpart o,ypart l);

93 fi

94 rrtitlepath := o--i{up}..{left}j--k{left}..{down}l--n;

95 if (rrtitlepos = "center"):

96 rrtitlestation := 0.5[n,o] shifted (0,rrtitleht/2);

97 elseif (rrtitlepos = "left"):

98 rrtitlestation := n shifted (rrborderrad/4,rrtitleht/2);

99 elseif (rrtitlepos = "right"):

100 rrtitlestation := o shifted (-rrborderrad/4,rrtitleht/2);

101 fi

102 elseif (rrtitleside = "bot"):

103 n := (xpart m,ypart f + (rrtitleht));

104 o := (xpart i,ypart f + (rrtitleht));

105 if (ypart n < ypart m):

106 n := (xpart n,ypart m);

107 o := (xpart o,ypart m);

108 fi

109 rrtitlepath := o--h{down}..{left}g--f{left}..{up}m--n;

110 if (rrtitlepos = "center"):

111 rrtitlestation := 0.5[n,o] shifted (0,-rrtitleht/2);

112 elseif (rrtitlepos = "left"):

113 rrtitlestation := n shifted (rrborderrad/4,-rrtitleht/2);

114 elseif (rrtitlepos = "right"):

115 rrtitlestation := o shifted (-rrborderrad/4,-rrtitleht/2);

116 fi

117 fi

118 rrfinalline = n--o;

119 picture rrtitlepic;

120 picture rrtitlepicb;

121 picture rrtitlebotborderpic;

122 rrtitlebotborderpic := image (draw rrfinalline

123 withcolor rrtitlebotbordercolor);

124 rrtitlepicb := image(draw rrtitlepath withcolor rrtitlebordercolor);

125 if (rrtitlebotborder = true):

126 addto rrtitlepicb also rrtitlebotborderpic;

127 fi

128 rrtitlepath := buildcycle(rrtitlepath,rrfinalline);

129 rrtitlepic := image(fill rrtitlepath withcolor rrtitlebgcolor);

130 if (rrtitlepos = "center"):

131 rrtitlelabel := image(label(TEX(rrtitlefont&" "&rrtitletext),

132 rrtitlestation) withcolor rrtitlecolor);

133 elseif (rrtitlepos = "left"):

134 rrtitlelabel := image(label.rt(TEX(rrtitlefont&" "&rrtitletext),

135 rrtitlestation) withcolor rrtitlecolor);

136 elseif (rrtitlepos = "right"):

137 rrtitlelabel := image(label.lft(TEX(rrtitlefont&" "&rrtitletext),

138 rrtitlestation) withcolor rrtitlecolor);

139 fi

140 if (rrbodyalign = "center"):

141 rrbodyaligncom := "\centering ";

142 elseif (rrbodyalign = "left"):

143 rrbodyaligncom := "\flushleft\vskip-\baselineskip ";

144 elseif (rrbodyalign = "right"):

145 rrbodyaligncom := "\flushright\vskip-\baselineskip ";

146 fi

147 picture rrbodytextpic;

148 if (rrtitleside = "top"):

149 rrbodytextpic :=

150 image(label.lrt(TEX("\parbox{"&decimal(rrwd-2rrborderwd-rrborderrad/2)&"bp}{"&rrbodyaligncom&rrtextfont&rrbodytext&"}"),n) withcolor rrtextcolor) shifted (rrborderrad/4,0);

151 elseif (rrtitleside = "bot"):

152 rrbodytextpic :=

153 image(label.urt(TEX("\parbox{"&decimal(rrwd-2rrborderwd-rrborderrad/2)&"bp}{"&rrbodyaligncom&rrtextfont&rrbodytext&"}"),n) withcolor rrtextcolor) shifted (rrborderrad/4,0);

154 fi

155 addto name also rrtitlepic;

156 addto name also rrtitlepicb;

157 addto name also rrtitlelabel;

158 addto name also rrbodytextpic;

159 fi

160 enddef;

X


